
pkgname=dialog
pkgver=1.3_20180621
_pkgver=1.3-20180621
pkgrel=1
epoch=1
pkgdesc="A tool to display dialog boxes from shell scripts"
arch=('x86_64')
url="https://invisible-island.net/dialog/"
license=('LGPL2.1')
depends=('ncurses' 'sh')
# version at http://invisible-island.net/dialog/CHANGES.html
source=("https://invisible-mirror.net/archives/dialog/${pkgname}-${_pkgver}.tgz")
md5sums=('1dba09cf45ac4ba40b3cb4a3f1d694b2')

build() {
  cd ${pkgname}-${_pkgver}
  
  ./configure --prefix=/usr \
    --mandir=/usr/share/man \
    --with-ncursesw \
    --enable-nls
  make 
}

package() {
  cd ${pkgname}-${_pkgver}
  
  make DESTDIR=${pkgdir} install install-lib
}
